

HELLO BUENOS DIAS GODDAG BONJOUR GUTEN TAG A\$="HELLO":PRINT A\$

#### READERS--

We are pleased to present the first issue of ColorCue, a newsletter for and about you and your personal computer, the COMPUCOLOR II. A newsletter such as this can serve many purposes, and as it begins to take form over the next few months, your contributions and criticisms will determine its character.

This publication provides a means of communication between us and COMPUCOLOR II owners, and, as things progress, between the owners themselves. We plan to offer explanations of various programming techniques that will help the new user become more adept. In addition, we will discuss features and technical aspects of the COMPUCOLOR II which will stimulate and inform even the most sophisticated owners/users.

In order to serve your needs, it is necessary that you make those needs known to us. We welcome any requests you may have for articles. If you're curious or perplexed about a certain aspect of programming or machine use, chances are that another user is too, and sufficient reader interest will generate a feature in ColorCue. Your suggestions and encouragements are also welcome and we plan to share some of them with other readers.

This paper will keep you abreast of innovations at Compucolor Corporation. We will be reporting on new options, peripherals, and

\*\*\*\*\*

#### This Issue's MENU:

1. Editor's Letter..... 1
2. Advanced Application  
Scrolling Patch ..... 2
3. Keeping It Simple  
Random Rectangles..... 4
4. Software Explained  
Star Trek Strategy..... 6
5. Users' Software Files..... 8
6. Coming Up ..... 8

\*\*\*\*\*

Library Diskettes as they are made available.

Many of you are eagerly awaiting the next Library Albums. Several new ones, including BlackJack and Personal Finance, will soon be available. If you have any suggestions for programs you would like to see, please let us know. Perhaps a favorite game or financial function would be a useful addition to the COMPUCOLOR II Program Library.

Many programmers, beginners and seasoned pros alike, have questions about the features of the COMPUCOLOR II, such as its use as a terminal, its particular form of BASIC, or its hardware specifications. Your questions have answers in the COMPUCOLOR II Programming and Reference Manual. It is a ready source of information for all who use the machine. It can be purchased through your dealer or ordered directly from us at Compucolor Corporation.



## ADVANCED APPLICATION

### How to Use The Scrolling Patch

Selective scrolling on the screen of the COMPUCOLOR II can be done quite easily by including a short BASIC routine in your own programs. This routine allows you to scroll a certain portion of the screen independently of the rest of the screen. It is used in several of the COMPUCOLOR programs in the Diskette Library, including STAR TREK. Because the routine is complicated, space prohibits our describing the details of how it works. Instead, we will show you how to use it.

First, initialize BASIC and key in the lines 0, 1, and 64000 through 65410 as listed below. If you plan to use the scrolling routine in a number of your programs, save the partially completed routine on disk so that you do not have to re-enter it each time that you want to use it. To save it, type SAVE"SCROLL" which saves the routine in a file called SCROLL.BAS on your diskette.

```
0 GOTO 65000
1 REM THERE MUST BE A LINE NUMBERED 1:STRING SPACE=25 BYTES
64000 GOSUB 65410
64010 Z=28672+128*Y+X+X:AD=TM+2: GOSUB 65400
64015 Z=128-W-W:AD=TM+25:GOSUB 65400
64020 POKE TM+5,H-1:POKE TM+7,W*(C+1):POKE TM+19,35*(1-C)
64030 RETURN

65000 GOSUB 65410:RESTORE 65030
65030 DATA 33,-1,-1,6,-1,14,-1,17,128,0,25,126
65040 DATA 17,128,255,25,119,35,-1,13,194,-1,-1
65050 DATA 17,-1,-1,25,5,194,-1,-1,201
65055 IF TM>65503 THEN TM=TM-32:GOTO 65080
65060 FOR I=1 TO 32:READ A
65065 IF A>=0 AND A<> PEEK (TM+1) THEN I = 32:TM=TM-32
65070 NEXT
65080 RESTORE 65030
65090 FOR I= 1 TO 32:READ A: POKE TM+I,A-(A<0):NEXT
65100 Z=TM+1: AD=33283:GOSUB 65400
65110 Z=TM:AD=ER:GOSUB 65400
65120 Z=TM+6:AD=TM+30:GOSUB 65400
65130 Z=TM+8:AD=TM+22:GOSUB 65400
65190 CLEAR 25: GOTO 1
65400 ZZ=INT (Z/256):POKE AD,Z-256*ZZ:POKE AD+1,ZZ:RETURN
65410 ER=32940:TM=256*PEEK(ER+1)+PEEK(ER):RETURN
```

To use the scrolling routine, you must specify a number of parameters that define the size and position of the area on the screen that will be scrolled. The 4 determinants are:

X - The x cursor position (0-63) of the upper left hand corner of the area to scroll

Y - The y cursor position (0-31) of the upper left hand corner of the area to scroll

H - The number of lines (>1) in the area to scroll

W - The number of characters (1-64) in each line in the scrolling area

A fifth parameter is necessary in order to specify whether or not the color status information is to scroll. It is defined as follows:

C - If C=0, then the color information does not scroll. If C=1, the color information is scrolled along with the character.

For example, if an 8 line by 16 character area is to be scrolled in the upper right hand corner of the screen, the parameters should be set as follows:

X=48: Y=0: H=8: W=16: C=1

By executing a GOSUB 64000, these parameters will be set into the assembly language scrolling routine. Thus, whenever a CALL(0) function is executed, the area will scroll up one line. The bottom line of the scrolled area is not spontaneously erased and must be erased by a PRINT statement. A completed scrolling routine should be similar in form and appearance to the one shown below. Lines 10 through 140 are a sample routine.

```
0 GOTO 65000
1 CLEAR 200: REM 200 BYTES OF STRINGS
10 REM SET SCROLLING AREA TO MIDDLE OF SCREEN
20 X=16: Y=8: H=16: W=32: C=1: GOSUB 64000
80 PLOT 27,24,6,32,12: REM SET PAGE MODE, COLOR, ERASE PAGE
90 PLOT 6,6
100 FOR LN= 1 TO 50
110 GOSUB 1000:PRINT "THIS IS LINE" LN
120 NEXT LN
130 END

1000 X=CALL(0)
1010 PLOT 3, X, Y+H-1: PRINT SPC(W)""
1020 PLOT 3,X,Y+H-1
1030 RETURN
```

The area will continue to scroll one line at a time each time line 1000 is executed. Several portions of the screen cannot be scrolled simultaneously, but they can be scrolled one after the other by executing GOSUB 64000 with the appropriate parameters before issuing each CALL(0) function.



## KEEPING IT SIMPLE

### Random Rectangles

The following program and its accompanying description are designed to make the new user/programmer feel more comfortable with the PLOT statement. This and other statements are fully described in our easy-to-understand Programming and Reference Manual, available through your dealer or direct from Compucolor Corporation. The Manual details the features of the machine and of the programming language BASIC. It starts out at an introductory level and progresses to sophisticated explanations of machine capabilities. It makes a handy reference, and when combined with study from one of the books recommended in the Instruction Manual, it will have the new user programming in no time at all.

The program below will draw randomly-sized rectangles of random color on the screen. The program is quite simple and should be easy to follow with but a little concentration.

The first step is the defining of a function that allows us to specify a range and have random numbers generated in that range (line 20). After having set the background color to black and cleared the screen (line 110) a loop is created with a FOR statement that will cause the program to repeat the drawing of rectangles a random number of times from 6 to 12 (line 210). Note the use of the user-defined function to select the number between 6 and 12.

The next step of the program is the random selection of the rectangle color. The colors have PLOT numbers from 0 to 7, but since a black rectangle would not show up on the pre-set black background, our random number function is set to choose only numbers from 1 to 7.

After the screen is prepared and the color set, the points that define the rectangle itself must be determined. As you can see in Figure 1 below, the four required coordinates can be derived from only 2 points, (X1,Y1) and (X2,Y2). Since the COMPUCOLOR II has 128 plot positions (from 0 to 127), the random function is set at 128 to determine coordinates for the two key points (lines 260, 270, 280, 290).




FIGURE 1

A subroutine is used to actually draw each rectangle on the screen. It is entered through line 300 with a GOSUB statement. The subroutine itself consists of entering the Point Plot mode by the PLOT 2 statement, and drawing the first point specified. The graphic Plot Submode 242, or Vector mode, is then used to draw the four sides of the rectangle. In this mode the computer draws a line from the last point plotted to the next specified point. Before using the mode, it is necessary to draw the first point in Point Plot mode (line 1000) to give the machine a starting point from which to draw the vector. After the initial point and the four

vectors are drawn, we escape from plot mode by the PLOT 255 command (line 1030) and control is then passed back to the main body of the program at line 400. This NEXT statement causes the program to go back and repeat the process in order to draw another of the 6 to 12 rectangles specified. When the program falls through the loop, that is, when all of the rectangles specified by the random number function in line 210 have been drawn, the program will be at line 500. This line has no actual computational function; it is inserted simply to cause a slight delay in the program before it reaches line 510 and repeats the entire process of screen clearing, choosing the number of rectangles, color, etc. Try entering the program into your COMPUCOLOR II and see if you get the results you expect.

```

10 REM DRAW RANDOM RECTANGLES
20 DEF FNR(X) = INT(X*RND(1))
100 REM SET COLOR AND ERASE PAGE
110 PLOT 6,0,12
200 REM DRAW 6 TO 12 RANDOM RECTANGLES
210 FOR I=1 TO 6+FNR(7)
220 REM SET VISIBLE FOREGROUND COLOR
230 PLOT 6,1+FNR(7)
250 REM COMPUTE RANDOM COORDINATES FOR RECTANGLE
260 X1 = FNR(128)
270 X2 = FNR(128)
280 Y1 = FNR(128)
290 Y2 = FNR(128)
300 GOSUB 1000:REM DRAW
400 NEXT I
500 FOR I=1 TO 1000: NEXT I
510 GOTO 100

1000 PLOT 2,X1,Y1
1010 PLOT 242,X2,Y1, X2,Y2, X1,Y2, X1,Y1
1020 PLOT 255
1030 RETURN

```

Notice that a little planning ahead has gone a long way here. The simple function created in line 20 is used a total of 6 times and saves a lot of steps. The program can be easily changed by specifying different ranges for color (line 230), number of rectangles (line 210), and size of rectangles (lines 260-290). You might try some variations. Remember that typing LINEFEED (↓) will halt program execution and return you to BASIC, and the program will still be in memory.

## SOFTWARE EXPLAINED

### Star Trek Strategy

The game Star Trek has become very popular, and the version created for the COMPUCOLOR II is among the most exciting ever played. The colors and graphics make it come alive for you, the user, Captain of the Enterprise. The game is easy to play and has a simple strategy, but because it is quite unlike any board games you might have played, it does take some getting used to. The first information you are asked to supply is the Star Date, and as you play several games you will see that the degree of difficulty of the game you get is a result of the Star Date you select. So, if you play a game and get destroyed, but you think you have a new strategy, just enter the star date of that game, and the same number of Klingons, Star Bases, and stars will be allotted at the start of the game. Some players have favorite games that they play over and over. If you really like a challenge--try Star Date 5!

The Galaxy is divided into an 8 by 8 quadrant grid which is in turn divided into an 8 by 8 sector grid. The position of the Enterprise is graphically shown on the short-range sensors, and its coordinates appear in large yellow characters in the screen's upper left corner.

The long-range sensors display the position of the Enterprise in a blinking number. The other numbers on the sensors give information about stars, (the number of which in any quadrant is displayed in yellow) Klingons, (displayed in red) and star bases (number of stars displayed in green). The contents of all adjacent quadrants are also visible, and as the Enterprise moves throughout the galaxy, the quadrants that come into range are displayed on the sensor scan. Eventually, this gives a clear picture of the galaxy, and you know the exact location of all Klingons, stars, and Star Bases.

It is important, as you have no doubt discovered during play, to learn the location of any starbases as soon as possible in the game. When the Enterprise comes within 1 sector of a Star Base, the shields are dropped, the Condition is "Docked", and all energy is fully restored. For this reason, many players prefer to avoid Klingons completely until a Starbase is located and they have a ready source of energy.

You will notice that when the game first starts, the status (appearing at the right of the screen), is CONDITION RED. This danger signal is given because there is no energy in the shields, and any Klingon attack would destroy the Enterprise. The first move of every game should be to put some units of energy in the shields. You will discover by trial and error the amount of energy needed to withstand the usual Klingon attack. Putting too much energy in the shields will severely limit your mobility, while putting in too little will leave you vulnerable to Klingon attack. Once the shields are prepped, you are ready to travel the universe in search of Klingons. Your ultimate goal, of course, is to rid the Galaxy of the evil Klingons, and control the area for the Federation.

The commands listed in the lower right of the screen are all that are needed to play the game. They are:

### 0-Warp Engine Control

Course is in a circular numerical vector arrangement as shown in Figure 2. Integer and real values may be used. Therefore, course 1.5 is halfway between 1 and 2. A vector of 9 is automatically changed to 1. After choosing course, warp speed is selected. A warp factor of 1 is the size of one quadrant. Therefore, to get from quadrant 6,5 to 5,5 you would use course 3, warp factor 1.

### 1-Shield Control

Defines the number of energy units to be assigned to the shields. The energy is taken from the total ship's energy.


### 2-Phaser Control

Allows you to destroy the Klingons by hitting them with suitably large numbers of energy units to deplete their power. The phasers are wide-range weapons, and require no target calculations.

### 3-Photon Torpedo Control

Torpedos must be precisely aimed using the same course selection described in command 0. Note the fate of Klingons after a direct hit!

The movement in the Star Trek game must be carefully planned because it is confusing at first. You use the ship's computer to lay in your course, and even though the distance and direction are mathematically calculated, you must determine the course yourself, and unless you stay ahead of the computer you're liable to end up lost in space! The course is computed as shown below:



There are a few details about which you should know in order to become an effective starship Captain. The first is that the status of devices, as shown in the lower right of the screen, can greatly affect your odds in any particular game. If the sensors go down, the displays will not be updated, and Klingons can move into your quadrant unannounced. It is important to be very careful to navigate AROUND stars. Attempting to go through a star that is visible on your short-range sensor screen will stop the ship at the point where it encounters the star. If a Warp Engine shut down appears on the Device Status record, your speed will be severely curtailed. It will take the elapse of several plays for Scotty to fix the engines, but eventually they will again be operable.

Klingons do not have to remain in the quadrant where you first encounter them. Each time you issue a command, Klingons within a distance of 3 quadrants may converge on your quadrant up to a maximum of 9 Klingons. It will take some practice to become truly adept at this challenging game, but with a little cleverness and some patience you will soon be a skilled player.

#### USERS' SOFTWARE FILES

##### Software Exchange

ColorCue will provide the means through which readers may sell, trade, or contribute programs to Compucolor Corporation. The next issue will feature a complete explanation of buying policies and standards for acceptance.

\*\*\*\*\*

##### COMING UP

Othello: COMPUCOLOR II vs. all challengers

User's Software Files: Software sales and acceptance policies

RS232-C Interface: Plugging in to the Big Machines

More PLOT's and some TIMELY Applications

\*\*\*\*\*

We hope you have found this first issue of ColorCue illuminating. In order that we meet your needs and interests, write to us with questions, comments and criticisms. Keep on experimenting and let us know what you come up with! Write to:

COMPUCOLOR CORPORATION  
P.O. BOX 569  
NORCROSS, GEORGIA 30071

ATTN: Susan G. Sheridan